Newsgroup: it.comp.www.asp
Titolo: Problemi di paginazione

Ciao a tutti,
sto provando a fare una paginazione, usando le solite istruzioni.
PageSize, PageCount e AbsolutePage, ma ricevo il seguente errore:
Tipo di errore:
ADODB.Recordset (0x800A0CB3)
Il set di record corrente non supporta i segnalibri. Potrebbe trattarsi di
una limitazione del provider o del tipo di cursore selezionato.
credits.asp, line 22
Come risolvo? Grazie mille!
--
Roberto Petillo

 ha scritto:

Quando apri il recordset, imposti il tipo di cursore?
rs.Open "stringa sql",connessione,1
Se puoi postare un po' di codice...
Ciao
-- 
AZ [Microsoft - .NET MVP]
Mia Home page: 
http://ciclismo.sitiasp.it
Asp.Net community: 
http://www.aspitalia.com
Il mio blog: 
http://blogs.aspitalia.com/az

 Come risolvo? Grazie mille!
basta che setti il cursore come dynamic.
rs.CursorType = adOpenDynamic oppure rs.open ("Select * From Table",
connessione, adOpenDynamic)
al posto della variabile adOpenDynamic puoi mettere direttamente il valore 2

 PageSize, PageCount e AbsolutePage, ma ricevo il seguente errore:
.....
Shine Software
www.shinesoftware.it
Ciao Roberto,
        usa questa funzione ...che funziona molto bene....
Questa è la chiamata:
basta che gli passi i parametri corretti, e poi fa tutto da sola.
    Dim intPage
    intPage = 1
    If Not IsEmpty(Request("page")) Then
     If IsNumeric(Request("page")) Then
      intPage = CLng(Request("page"))
     End If
    End If
    Response.Write Paging(intPage, 25, 200)
-----------------
e questa la funzione:
<%
    Public Function Paging(ByVal intPage, ByVal intPageCount, ByVal
intRecordCount)
     Dim strQueryString
     Dim strScript
     Dim intStart
     Dim intEnd
     Dim strRet
     Dim i
     If intPage > intPageCount Then
      intPage = intPageCount
     ElseIf intPage < 1 Then
      intPage = 1
     End If
     If intRecordCount = 0 Then
      strRet = "No Records Found"
     ElseIf intPageCount = 1 Then
      strRet = "All Records Shown"
     Else
      For i = 1 To Request.QueryString.Count
       If LCase(Request.QueryString.Key(i)) <> "page" Then
        strQueryString = strQueryString & "&"
        strQueryString = strQueryString &
Server.URLEncode(Request.QueryString.Key(i)) & "="
        strQueryString = strQueryString &
Server.URLEncode(Request.QueryString.Item(i))
       End If
      Next
      For i = 1 To Request.Form.Count
       If LCase(Request.Form.Key(i)) <> "page" Then
        strQueryString = strQueryString & "&"
        strQueryString = strQueryString &
Server.URLEncode(Request.Form.Key(i)) & "="
        strQueryString = strQueryString &
Server.URLEncode(Request.Form.Item(i))
       End If
      Next
      If Len(strQueryString) <> 0 Then
       strQueryString = "?" & Mid(strQueryString, 2) & "&"
      Else
       strQueryString = "?"
      End If
      strScript = Request.ServerVariables("SCRIPT_NAME") & strQueryString
      If intPage <= 10 Then
       intStart = 1
      Else
       If (intPage Mod 10) = 0 Then
        intStart = intPage - 9
       Else
        intStart = intPage - (intPage Mod 10) + 1
       End If
      End If
      intEnd = intStart + 9
      If intEnd > intPageCount Then intEnd = intPageCount
      strRet = "Page " & intPage & " of " & intPageCount & ": "
      If intPage <> 1 Then
       strRet = strRet & "<a href=""" & strScript
       strRet = strRet & "page=" & intPage - 1
       strRet = strRet & """><<Prev</a> "
      End If
      For i = intStart To intEnd
       If i = intPage Then
        strRet = strRet & "<b>" & i & "</b> "
       Else
        strRet = strRet & "<a href=""" & strScript
        strRet = strRet & "page=" & i
        strRet = strRet & """>" & i & "</a>"
        If i <> intEnd Then strRet = strRet & " "
       End If
      Next
      If intPage <> intPageCount Then
       strRet = strRet & " <a href=""" & strScript
       strRet = strRet & "page=" & intPage + 1
       strRet = strRet & """>Next>></a> "
      End If
     End If
     Paging = strRet
    End Function
    %>